



**Excel nieuwe functies in Office 365 SCHAKELEN &  
ALS.VOORWAARDEN.docx**

**Handleiding van Helpmij.nl**

**Auteur: CorVerm**

**november 2017**

“ Dé grootste en gratis computerhelpdesk van Nederland ”

Helpmij.nl



SCHAKELEN en ALS.VOORWAARDEN zijn nieuwe functies in Excel (Office 365). Door de functie SCHAKELEN te gebruiken is het overbodig de functie ALS (soms vaak) te moeten nesten. Evenzo met de functie ALS.VOORWAARDEN. Zijn er voorwaarden verbonden aan het resultaat dat bereikt moet worden dan hoef je ook met deze functie een formule niet te nesten. Zo bereik je op een makkelijker en minder foutgevoelige manier je doel. Aan de hand van twee voorbeelden worden de functies uitgelegd.

### De functie SCHAKELEN

	A	B	C	D	E	F	G
1	<b>Datum</b>	<b>Weekdag</b>	<b>Dag</b>				
2	22-9-2017	weekdag	andere dag				andere dag
3	23-9-2017	weekend	zaterdag				
4	24-9-2017	weekend	zondag				
5	25-9-2017	weekdag	andere dag				
6	26-9-2017	weekdag	andere dag				
7	=SCHAKELEN(WEEKDAG(A2);1;"weekend";7;"weekend";"weekdag")						
8	=SCHAKELEN(WEEKDAG(A2);1;"zondag";7;"zaterdag";"andere dag")						
9							
10	=ALS(WEEKDAG(A2)=1;"weekend";ALS(WEEKDAG(A2)=7;"weekend";"andere dag"))						

Aan de hand van een aantal datums moet worden bepaald of het een doordeweekse dag, of zaterdag dan wel zondag (weekend) is. Met de functie SCHAKELEN is dat eenvoudig te bepalen. Met de formule: =SCHAKELEN(WEEKDAG(A2);1;"weekend";7;"weekend";"weekdag") is dat zo gepiept.

```
=SCHAKELEN(WEEKDAG(A2);1;"weekend";7;"weekend";"weekdag")
```

```
SCHAKELEN(expressie; value1; result1; [value2; result2]; [default_or_value3; result3]; [default_or_value4; r
```

Het eerste argument is *expressie*. Daarvoor kun je een functie gebruiken of een celwaarde opgeven. In het geval van het voorbeeld is de functie WEEKDAG gebruikt (genest). De andere argumenten zijn: *waarde1 (value1)* en dan *resultaat1 (result1)* enzovoorts. Raar maar waar zijn de argumenten voor de functie niet vertaald zijn in het Nederlands. De functie WEEKDAG haalt het dagnummer uit de datum (22-9-2017 heeft als weekdag 6). In de formule is opgegeven dat dag 1 "weekend" is, evenzo voor dag 7. Wordt niet aan die voorwaarden voldaan dan moet de uitkomst van de formule "weekdag" zijn. Zo simpel is dat. Je kunt tot 126 overeenkomende waarden en resultaten evalueren.

Met de ALS-functie moet je in dit geval (slechts) tweemaal nesten. De formule ziet er dan zo uit: =ALS(WEEKDAG(A2)=1;"weekend";ALS(WEEKDAG(A2)=7;"weekend";"andere dag")). De functie SCHAKELEN opent de formule met een haakje en gesloten met een haakje. In tegenstelling tot de ALS-functie, die moet afgesloten worden met net zoveel haakjes als er ALS-functie zijn genest.


De formule in kolom C is: =SCHAKELEN(WEEKDAG(A2);1;"zondag";7;"zaterdag";"andere dag").

## Nogmaals SCHAKELEN en ALS.VOORWAARDEN

	A	B	C	D	E	F
1	Pool 1	Pool 2	Pool 3	Kolomnr.	Schakelen	ALS.VW
2	12	10	9	1	12	goed
3	15	12	13		#N/B	#N/B
4	18	14	15	3	15	goed
5	21	16	18	2	16	uitstekend
6	24	10	21	2	10	goed
7	27	20	24	2	20	uitstekend
8	30	22	27	3	27	uitstekend
9						
10	=SCHAKELEN(D2;1;A2;2;B2;3;C2)					
11	=ALS(D3="";"";SCHAKELEN(D3;1;A3;2;B3;3;C3))					

In het overzicht zijn drie kolommen (A:C) gevuld met getallen. In kolom D zijn de kolomnummers ingevoerd en de bedoeling is dat, aan de hand van de kolomnummers, de uitkomst in kolom E tevoorschijn komt. Dat aan de hand van wederom de functie SCHAKELEN. De formule in E2 is dan: =SCHAKELEN(D2;1;A2;2;B2;3;C2). De formule is uiteraard naar beneden gekopieerd.

In dit geval is expressie een cel verwijzing, namelijk D2.



De eerste waarde (1) staat voor kolom 1 (A) en zo verder tot kolom 3 (C). Het resultaat zijn de cel verwijzingen. Door een kolomnummer in te voeren in kolom D is het resultaat van de formule in kolom E te zien.

Echter, in E3 staat de foutwaarde #N/B. Ook in F3 trouwens, maar die foutwaarde verdwijnt als de formule is aangepast in E3. Hoe dat komt? De functie SCHAKELEN heeft geen optie om een foutwaarde te ondervangen. Aangezien D3 leeg is geeft dat dus de foutwaarde. Er zal toch iets aan de formule moeten worden aangepast, en zal de ALS functie genest moeten worden. De formule wordt dan: =ALS(D3="";"";SCHAKELEN(D3;1;A3;2;B3;3;C3)), of: =ALS(D3="";"geen waarde";SCHAKELEN(D3;1;A3;2;B3;3;C3)). In het eerste geval blijft de cel leeg, in het andere geval komt er de tekst "geen waarde" (zonder aanhalingstekens) in de cel te staan.

	A	B	C	D	E	F
3	15	12	13	1	15	goed

Zoals je ziet is de foutwaarde nu verholpen

Zou de formule met de ALS-functie worden opgebouwd dan zou die er zo uitzien: =ALS(D2="";"";ALS(D2=1;A2;ALS(D2=2;B2;ALS(D2=3;C2)))) De functie SCHAKELEN is dus een stuk eenvoudiger.

## ALS.VOORWAARDEN

In kolom F wordt beoordeeld of de uitkomst in kolom E matig, goed of uitstekend is. Dat is gedaan met de functie ALS.VOORWAARDEN. De formule ziet er zo uit:

=ALS.VOORWAARDEN(E2<10;"matig";E2<=15;"goed";E2>15;"uitstekend"). Zou de formule met de

ALS-functie zijn opgebouwd dan ziet die er zo uit:

=ALS(D2="";"leeg";ALS(E2<10;"matig";ALS(E2<=15;"goed";ALS(E2>15;"uitstekend";"")))). Ook nu is het grote verschil te zien en het voordeel van de functie ALS.VOORWAARDEN bewezen. Trouwens, om foutwaarden in kolom F te voorkomen (uitgaande van F3) dan is de formule:

=ALS.VOORWAARDEN(D3="";"";E3<10;"matig";E3<=15;"goed";E3>15;"uitstekend")

	A	B	C	D	E	F
1	<b>Pool 1</b>	<b>Pool 2</b>	<b>Pool 3</b>	<b>Kolomnr.</b>	<b>Schakelen</b>	<b>ALS.VW</b>
2	12	10	9	2	10	goed
3	15	12	13	1	15	goed
4	18	14	8	3	8	matig
5	21	16	18	2	16	uitstekend
6	24	10	21	2	10	goed
7	27	20	24	2	20	uitstekend
8	5	22	27	1	5	matig

Ik wil nogmaals benadrukken dat de nieuwe functies alleen in Office 365 te vinden zijn. Dus niet in Office 2016.

Volgende maand komen de functies TEKST.COMBINEREN en MAX.ALS.VOORWAARDEN aan de beurt.

Helpmij.nl